/******************************************************************************/
// $Date: 2016-06-03 10:01:35 +0200 (Fri, 03 Jun 2016) $
// $Author: feron $
// $Rev: 15991 $
/******************************************************************************/
module SideLinkUE_ASP_TypeDefs {

  /* Type definitions for configuration of UE sidelink simulator SS-UE;
   * Common design principles:
   * Semantics of OMIT: for all TTCN-3 type definitions used in ASPs omit means "keep as it is" =>
   * - on initial configuration in general all fields shall be provided
   * - no default values for fields are foreseen
   * - if necessary non-existence of information shall be explicitly configured
   *   (e.g. with a union of "no configuration" and "configuration parameters"
   * - fields within structures imported from the core spec are excepted from this rule */
  import from EUTRA_RRC_ASN1_Definitions language "ASN.1:2002" all;
  import from PC5_RRC_Definitions language "ASN.1:2002" all;
  import from EUTRA_SideLink_Preconf language "ASN.1:2002" all;
  import from CommonDefs all;
  import from EUTRA_CommonDefs all;
  import from EUTRA_ASP_TypeDefs all;
  import from EUTRA_ASP_DrbDefs all;
  import from PC5_Message_TypeDefs all;
  
  //===========================================================================================
  // DATA ASPs
  //===========================================================================================
  group SideLinkUE_Data {
  
  //****************************************************************************
  // Data ASP definitions
  //****************************************************************************
  type record SL_DATA_REQ {                /* Common ASP to send PDUs on PC5 channels SL-DCH or SLRB */
    SL_ReqAspCommonPart_Type    Common,    /* SS_UE_Id : identifier of the SL UE
                                            * RoutingInfo : SL-DCH or SLRB identity
                                            * TimingInfo : For Direct Discovery see description in DiscPeriodResourceList_Type;
                                                           For communication: SFN/Subframe or DFN/DirectSubframe to use to start sending data PDUs.
                                            * ControlInfo : CnfFlag:=false; FollowOnFlag:=false */
    SL_U_PlaneTx_Type           U_Plane
  };
  
  type record SL_DATA_IND {                /* Common ASP to receive PDUs from PC5 channels SL-DCH or SLRB */
    SL_IndAspCommonPart_Type   Common,     /* SS_UE_Id : identifier of the SL UE
                                            * RoutingInfo : SL-DCH or SLRB identity
                                            * TimingInfo : SFN/Subframe or DFN/DirectSubframe when message has been received
                                            */
    SL_U_PlaneRx_Type          U_Plane
  };
  
  //****************************************************************************
  // U Data type definitions
  //****************************************************************************
  type record SL_U_PlaneTx_Type {
    SL_U_Plane_Type         U_Plane,
    SL_DataScheduling_Type  SL_DataScheduling  optional /* Applied on SL-DCH and PSCCH/PSSCH */
  };
  
  type record SL_U_PlaneRx_Type {
    SL_U_Plane_Type                 U_Plane,
    FreqDomainSchedulExplicit_Type  FreqDomainSchedul    optional /* Applied on SL-RB and SL-DCH*/
  };
  
  type union SL_U_Plane_Type {
    SL_DCH_Message_Type   Discovery,    /* SL-DCH message */
    PDCP_PDUList_Type     PdcpPdu,      /* SS configuration: PDCP no header removal - In Rx only one PDCP PDU is expected */
    PDCP_SDUList_Type     PdcpSdu       /* SS configuration: PDCP header removal - In Rx only one PDCP SDU is expected */
  };
  
  type union SL_DCH_Message_Type {    /* SL-DCH message */
    PC5_DISCOVERY     pC5_DISCOVERY
  };
  
  type union SL_DataScheduling_Type  {
    DiscPeriodResourceList_Type         DiscPeriodResource, /* Applied for SS-UE transmission on SL-DCH/PSDCH */
    CommPeriodResourceList_Type         CommPeriodResource  /* Applied for SS-UE transmission on PSCCH */
  };
  
  type record DiscPeriodResourceList_Type {
    /* Parameters for transmission of the PC5_DISCOVERY message during NumberOfTransmissionPeriod discovery Periods */
    /* TimingInfo in Common indicates the SFN/subframe of the first discovery period */
    integer                 NumberOfTransmissionPeriod, /* Number of consecutive discovery period */
    integer                 N_PDSCH                     /* Resource number in the range 0..(Nt*Nf-1) (see TS 36.213 cl 14.3.1)
                                                           Multiple PC5_DISCOVERY messages can be transmitted within a discovery period, different SL_DATA_REQ and N_PDSCH values will be provided
                                                         */
  };

  type record of CommPeriodResource_Type CommPeriodResourceList_Type;  /* One entry in the list per STCH packets provided in the U_Plane field
                                                                          One entry per communication period */

  type record CommPeriodResource_Type {
    /* Parameters for transmission of the STCH packets */
    /* TimingInfo in Common indicates the SFN/subframe of the first communication period */
    integer                 N_PSCCH       /* PSCCH Resource number (see TS 36.213 cl 14.2.1.1)
                                             Multiple STCH packets can be transmitted within a communication period, different SL_DATA_REQ and N_PSCCH values will be provided
                                           */
  };

  
  //****************************************************************************
  // Port definitions
  //****************************************************************************
  type port SL_DATA_PORT message {
    out SL_DATA_REQ;
    in  SL_DATA_IND;
  };
  
  } //End of SideLinkUE_Data
  
  //===========================================================================================
  // CONFIGURATION ASPs
  //===========================================================================================
  group SideLinkUE_Configuration {
   
  //****************************************************************************
  // SL Routing
  //****************************************************************************
  group SL_Routing_Timing {
  
  type enumerated SS_UE_Id_Type { /* Simulated UE-ID */
    ss_UE_NonSpecific,  /* Represents all Simulated UEs */
    ss_UE_Id1,
    ss_UE_Id2,
    ss_UE_Id3
  };
  
  type union SL_RoutingInfo_Type {
    Null_Type                     None,
    SL_ChannelRadioBearerId_Type  ChannelRadioBearerId
  };
  
  type union SL_ChannelRadioBearerId_Type {
    Null_Type   SLDCH,         // Discovery SL-DCH
    SLRB_Identity_Type   SLRB  // Communication SL-RB
  };
  
  type union SL_TimingInfo_Type {
    TimingInfo_Type            InCoverageTiming,    /* SFN/subframe*/
    SL_OoC_TimingInfo_Type     OutOfCoverageTiming  /* DFN/DirectSubframe */
  };
  
  type union SL_OoC_TimingInfo_Type { /* Out of coverage timing info*/
    DirectSubFrameTiming_Type   DirectSubFrame,
    Null_Type                   Now,    /* to be used in REQ ASPs when there is no 'activation time'  */
    Null_Type                   None    /* to be used in CNF ASPs but not for EnquireTiming */
  };
  
  type record  DirectSubFrameTiming_Type {
    DirectFrameNumber_Type    DFN,
    DirectSubframeNumber_Type DSubframe
  };
  
  type MasterInformationBlock_SL.directFrameNumber_r12      DirectFrameNumber_Type;
  type MasterInformationBlock_SL.directSubframeNumber_r12   DirectSubframeNumber_Type;
  type integer SLRB_Identity_Type;
  
  
  } // End of SL_Routing_Timing
  
  //****************************************************************************
  // SL System Request
  //****************************************************************************
  group SL_SystemRequestAsp {
  
  //----------------------------------------------------------------------------
  // REQUEST Common part
  //----------------------------------------------------------------------------
  group SL_RequestAspCommon_Part {
  
  type record SL_ReqAspCommonPart_Type {
    SS_UE_Id_Type           SS_UE_Id,    /* SS-UE ID */
    SL_RoutingInfo_Type     RoutingInfo, /* Channel or RB Id */
    SL_TimingInfo_Type      TimingInfo,  /* In coverage or Out of coverage timing */
    ReqAspControlInfo_Type  ControlInfo
  };
  
  //----------------------------------------------------------------------------
  // REQUEST Config part
  //----------------------------------------------------------------------------
  
  //-------------------------------------
  // Discovery and Communication common types
  type MasterInformationBlock_SL.sl_Bandwidth_r12 SL_Bandwidth_Type;
  
  type union SL_SystemRequest_Type {
    SS_UE_ConfigRequest_Type            SS_UE_Config,     /* Configure/release a SS-UE */
    Null_Type                           EnquireTiming,    /* Get SFN/subframe number or DFN/Direct subframe time for this SS-UE */
    SL_L1Mac_IndicationControl_Type     L1MacIndCtrl,     /* To configure SS_UE to generate indications for L1/MAC events */
    SL_Security_Type                    SL_Security       /* StartRestart/Release of SLRB security */
  };
  
  type record SL_L1Mac_IndicationControl_Type {
    /* NOTE:
       Initially all indications are disabled in SS (i.e. it shall not be necessary in 'normal' test cases to use this primitive but only if a specific indication is needed); omit means indication mode is not changed
    */
    L1Mac_IndicationMode_Type       DiscSync        optional,  /* To enable/disable reporting of start and stop of reception of Discovery SLSS as configured.
                                                                * NOTE:
                                                                * this is applicable when SS_UE is configured with RxDiscSyncRef
                                                                * SS reports an indication every time when SLSS reception from UE has toggled in the subframes configured for SLSS reception
                                                                */
    L1Mac_IndicationMode_Type       CommSLSS_MIBSL  optional   /* To enable/disable reporting of start and stop of reception of communication SLSS and MIB-SL as configured.
                                                                * NOTE:
                                                                * this is applicable when SS_UE is configured with CommRxSyncRef
                                                                * SS reports an indication every time when SLSS/MIB-SL reception from UE has toggled in the subframes configured for SLSS/MIB-SL reception
                                                                */
  };
  
  type union SS_UE_ConfigRequest_Type {
    SS_UE_ConfigInfo_Type       AddOrReconfigure,   /* For one SS-UE configuration:
                                                     * SS_UE_Id : identifier of the SS-UE to be configured
                                                     * RoutingInfo : None
                                                     * TimingInfo : Now (for initial configuration)
                                                     * ControlInfo : CnfFlag:=true; FollowOnFlag:=false (in general) */
    Null_Type                   Release             /* To remove all SS-UEs completely
                                                     * SS_UE_Id : ss_UE_NonSpecific
                                                     * RoutingInfo : None
                                                     * TimingInfo : Now
                                                     * ControlInfo : CnfFlag:=true; FollowOnFlag:=false (in general) */
  };

  type record SS_UE_ConfigInfo_Type {
    SS_UE_ProSeCoverage_Type    SS_UE_ProSeCoverage optional,  /* Mandatory in the first configuration - omit means no change
                                                                * InCoverage for discovery
                                                                */
    SS_UE_Power_Type            Power        optional,         /* Reference power for the RS of each antenna in Tx
                                                                * Initially configured as switched ON
                                                                */
    DiscConfig_Type             DiscConfig optional,
    CommConfig_Type             CommConfig optional
  };
  
  type record SS_UE_Power_Type {
    AbsoluteCellPower_Type      MaxReferencePower optional,      /* Maximum value of SS-UE reference power (in dBm/15kHz as per TS 36.508, clause 4.3.4.1);
                                                                  * SS-UE is initialised with this reference power;
                                                                  * its value can't be reconfigured during test execution, attenuation is used to change SS-UE power level
                                                                  * its value is the upper bound of the SS-UE power during the test case */
    Attenuation_Type            Attenuation                      /* Initial attenuation: ON */
  };
  
  type record OutOfCoverageConfig_Type {
    ARFCN_ValueEUTRA_r9         CarrierFreq_r12,
    SL_Bandwidth_Type           SL_bandwidth_r12,
    TDD_ConfigSL_r12            TDD_ConfigSL_r12,
    SS_UE_TimingInfo_Type       SS_UE_TimingInfo
  };
  
  type record SS_UE_TimingInfo_Type {  /* SS_UE Timing */
    integer DSubframeOffset (0..307199),   /* frame duration Tf = 307200 * Ts = 10ms; System Time Unit Ts = 1/(15000 * 2048) */
    integer DfnOffset (0..1023)            /* (assuming 10 bit DFN) */
  };

  type record InCoverageConfig_Type {
[bookmark: _GoBack]    ARFCN_ValueEUTRA_r9         UL_CarrierFreq,
    Ul_Bandwidth_Type           UL_bandwidth,
    TDD_Config                  TDD_Config optional,  /* omitted for FDD */
    CellTimingInfo_Type         CellTimingInfo
  };
  
  type union SS_UE_ProSeCoverage_Type {
    /* For Discovery only InCoverage is applied */
    EUTRA_CellIdInCoverageConfig_Type       InCoverage,    /* In coverage SS-UE shall use Band/Bandwidth/UL frequency/Cell timing/TDD Frame structure/CP of the associated EUTRA cell */
    OutOfCoverageConfig_Type    OutOfCoverage
  };

  //-------------------------------
  // Physical channels
  
  type record PrimarySideLinkSyncSignal_Type {
    SideLinkChannelPower_Type       RelativeTxPower       optional   /* Power for PSSS relative to the SS-UE power level - value in dB */
  };
  
  type record SecondarySideLinkSyncSignal_Type {
    SideLinkChannelPower_Type       RelativeTxPower       optional  /* Power for SSSS relative to the SS-UE power level - value in dB */
  };
  
  } // End of SL_RequestAspCommon_Part
 
  //----------------------------------------------------------------------------
  // Discovery specific types
  //----------------------------------------------------------------------------
  group Discovery_Specific {

  type record DiscConfig_Type {
    /* Acc. to 36.321: HARQ configured with fixed TBS=232 - MAC configured in TM
     * PC5_DISCOVERY msg = 232 bits, transmitted in one MAC TM PDU.
     * The PSDCH is transmitted on two contiguous physical resource blocks per slot.
     * The SideLink process is configured with a maximum number of HARQ retransmissions by RRC: numRetx.
     */
    SL_DCH_TxConfig_Type    SL_DCH_Tx      optional, /* Corresponding to UE Rx resource pool */
    SL_DCH_RxConfig_Type    SL_DCH_Rx      optional  /* Corresponding to UE Tx resource allocation*/
  };
  
  type union SL_DCH_TxConfig_Type {
    SL_DCH_TxConfigInfo_Type    AddOrReconfigure,
    Null_Type                   Release
  };

  type record SL_DCH_TxConfigInfo_Type {
    DiscResourcePool_Type     SS_UE_Tx  optional,
    DiscTxSyncRefConfig_Type  DiscTxSyncRef  optional, /* Configure SS_UE to transmit Sync SLSS with PC5_DISCOVERY Tx
                                                          No prose yet - See 36.331 clause 5.10.5 Note 4
                                                        */
    PsdchConfig_Type          Psdch  optional
  };
  
  type union SL_DCH_RxConfig_Type {
    SL_DCH_RxConfigInfo_Type    AddOrReconfigure,
    Null_Type                   Release
  };
  
  type record SL_DCH_RxConfigInfo_Type {
    DiscResourcePoolList_Type   SS_UE_Rx  optional,
    DiscRxSyncRefConfig_Type    DiscRxSyncRef  optional  /* For SS_UE to detect SLSS transmitted by UE under test
                                                            It is assumed that only a single Rx resource pool will be specified for the receipt of SLSS by SS-UE
                                                         */
  };
  
  type SL_DiscResourcePool_r12.discPeriod_r12 DiscPeriod_r12_Type;
  type SL_DiscResourcePool_r12.numRetx_r12 DiscNumRetx_r12_Type;
  type SL_DiscResourcePool_r12.numRepetition_r12 DiscNumRepetition_r12_Type;
    
  type record DiscResourcePool_Type  {
    /* All fields are mandatory for initial configuration; omit means "keep as it is" */
    SL_CP_Len_r12               cp_Len_r12              optional,
    DiscPeriod_r12_Type         discPeriod_r12          optional,
    DiscNumRetx_r12_Type        numRetx_r12             optional,
    DiscNumRepetition_r12_Type  numRepetition_r12       optional,
    SL_TF_ResourceConfig_r12    tf_ResourceConfig_r12   optional,
    SL_TF_IndexPairList_r12b    discTF_IndexList_r12b   optional /* Applied in SL-DCH Rx when UE is configured in transmission mode 2B
                                                                    To ensure SS-UE and UE are timing aligned, the indexes are provided to SS-UE using the same activation as the one used to send the signalling reconfiguration message to UE
                                                                 */
  };
  
  type record length(1..maxSL_TxPool_r12) of DiscResourcePool_Type DiscResourcePoolList_Type;
  
  type union DiscTxSyncRefConfig_Type {
    /* To start/reconfigure or stop transmitting SLSS (irrespective of any measurement) */
    DiscTxSyncRefConfigInfo_Type    AddOrReconfigure,
    Null_Type                       Release
  };
  
  type union DiscRxSyncRefConfig_Type {
    DiscSyncConfigParam_Type  AddOrReconfigure,
    Null_Type                 Release
  };
  
  type record DiscTxSyncRefConfigInfo_Type {
    /* All fields are mandatory for initial configuration; omit means "keep as it is" */
    DiscSyncConfigParam_Type           SyncConfig  optional,
    PrimarySideLinkSyncSignal_Type     Psss        optional,
    SecondarySideLinkSyncSignal_Type   Ssss        optional
  };
  
  type record DiscSyncConfigParam_Type {
    /* All fields are mandatory for initial configuration; omit means "keep as it is" */
    SL_CP_Len_r12               syncCP_Len_r12  optional,
    SL_OffsetIndicatorSync_r12  syncOffsetIndicator_r12 optional,
    SLSSID_r12                  slssid_r12  optional
  };
  
 type integer SideLinkChannelPower_Type;  /* see TS 36.213, clause 14 */
  
  type record PsdchConfig_Type {
    SideLinkChannelPower_Type      RelativeTxPower  optional /* Power for PSDCH relative to the SS-UE power level - value in dB */
  };
  
  } // End of Discovery_Specific
  
  //----------------------------------------------------------------------------
  // Communication specific types
  //----------------------------------------------------------------------------
  group Communication_Specific {
  
  type record CommConfig_Type {
    CommSyncConfig_Type  CommSyncRef     optional,
    CommRxConfig_Type    CommRxConfig    optional,
    CommTxConfig_Type    CommTxConfig    optional
  };
  
  //---------------------------------------
  // SyncRef Communication specific types
  type record CommSyncConfig_Type {
    CommTxSyncRefConfig_Type      CommTxSyncRef optional,
    CommRxSyncRefConfig_Type      CommRxSyncRef optional,
    SL_SyncRef_Type               SynchronisationSource optional /* Applied when OoC - When In coverage: omit
                                                                  * ueSyncRef
                                                                  *    UE under test is SyncRef UE
                                                                  *    SS_UE shall synchronise from received SLSS/MIB-SL configured in CommRxSyncRef
                                                                  *    UE shall be transmitting SL-SS when SS_UE is switched on
                                                                  * sS_UE_SyncRef
                                                                  *    SS_UE is SyncRef UE
                                                                  *    SS_UE shall initiate DFN/DirectSubrame and transmit SLSS/MIB-SL acc to CommTxSyncRef
                                                                  */
  };
  
  type union CommTxSyncRefConfig_Type {
    /* To start/reconfigure or stop transmitting SLSS/MIB-SL (irrespective of any measurement) */
    CommTxSyncRef_Type          AddOrReconfigure,  /* Add/re-configure SLSS or SBCCH  */
    Null_Type                   Release            /* Stop SLSS and SBCCH transmission */
  };
  
  type union CommRxSyncRefConfig_Type {
    CommRxSyncRef_Type          AddOrReconfigure,  /* Add/re-configure reception of SLSS or SBCCH */
    Null_Type                   Release            /* Release/stop reception of SLSS and SBCCH transmission */
  };
    
  type record CommRxSyncRef_Type {
    SL_CP_Len_r12               SyncCP_Len_r12,
    SLSSID_r12                  Slssid_r12 optional,      /* Omit means any value is expected */
    SL_OffsetIndicatorSync_r12  SyncOffsetIndicator1_r12,
    SL_OffsetIndicatorSync_r12  SyncOffsetIndicator2_r12 optional /* When present both SyncOffsetIndicator1_r12 and SyncOffsetIndicator2_r12 shall ne monitored */
  };
  
  type record CommTxSyncRef_Type {
    /*
      SBCCH - RLC TM - MAC TM
      SS will perform padding, if required due to the granularity of the TF signalling, as defined in 36.331 cl 8.5.
    */
    CommTxSyncRefParam_Type             TxSyncConfig    optional,
    MasterInformationBlock_SL           MIB_SL          optional, /* DFN and Direct subframe set to a dummy value - to be set by SS-UE */
    PrimarySideLinkSyncSignal_Type      Psss            optional,
    SecondarySideLinkSyncSignal_Type    Ssss            optional,
    PsbchConfig_Type                    Psbch           optional  /* The transmit power of PSBCH is same as the transmit power of primary sidelink synchronisation signal */
  };
  
  type record CommTxSyncRefParam_Type {
    SL_CP_Len_r12               SyncCP_Len_r12,
    SLSSID_r12                  Slssid_r12 optional,
    SL_OffsetIndicatorSync_r12  SyncOffsetIndicator1_r12
  };
  
  type enumerated SL_SyncRef_Type {sS_UE_SyncRef, ueSyncRef};

  //---------------------------------------
  // Tx/Rx Communication specific types
  
  type union CommRxConfig_Type {
    /* SS-UE ID : identifier of the simulated UE being configured
     * RoutingInfo : none
     * TimingInfo : 'Now' in common cases
     * ControlInfo : CnfFlag:=true; FollowOnFlag:=false (in general)
     */
    CommRx_ConfigInfo_Type        AddOrReconfigure, /* Add/re-configure SideLink PSCCH/PSSCH/SL-SCH/STCH/RLC/PDCP
                                                     */
    Null_Type                     Release           /* Release SideLink PSCCH/PSSCH/SL-SCH/STCH/RLC/PDCP
                                                     */
  };
  
  type union CommTxConfig_Type {
  /*
   * SS-UE ID : identifier of the simulated UE being configured
   * RoutingInfo : none
   * TimingInfo : 'Now' in common cases
   * ControlInfo : CnfFlag:=true; FollowOnFlag:=false (in general)
   */
    CommTx_ConfigInfo_Type        AddOrReconfigure, /* Add/re-configure SideLink PSCCH/PSSCH/SL-SCH/STCH/RLC/PDCP
                                                     */
    Null_Type                     Release           /* Release  SideLink PSCCH/PSSCH/SL-SCH/STCH/RLC/PDCP
                                                     */
  };
  
  type record CommRx_ConfigInfo_Type {
    // It is assume that one SSUE layer 2 ID/one SLRB needs to be handled
    CommResourcePoolList_Type  ResourcePoolList  optional,  /* Configure subframes reserved for PSSCH and PSCCH
                                                             * Present for first configuration
                                                             * omit means no change
                                                             */
    SL_SCH_Config_Type     SL_SCH_Config    optional,       /* SL-SCH/MAC configuration
                                                             * Present for first configuration
                                                             * omit means no change
                                                             */
    STCH_ConfigList_Type   STCH_ConfigList  optional
  };
  
  type record length(1..maxSL_TxPool_r12) of CommResourcePool_Type CommResourcePoolList_Type;

  type SL_CommConfig_r12.commTxResources_r12.setup.scheduled_r12.mcs_r12 Mcs_Type;
  
  type record CommResourcePool_Type {
    SL_CP_Len_r12               sc_CP_Len_r12 optional,
    SL_PeriodComm_r12           sc_Period_r12 optional,
    SL_TF_ResourceConfig_r12    sc_TF_ResourceConfig_r12 optional,
    SL_CP_Len_r12               data_CP_Len_r12 optional,
    SL_HoppingConfigComm_r12    dataHoppingConfig_r12 optional,
    SL_TF_ResourceConfig_r12    data_TF_ResourceConfig_r12 optional,
    SL_TRPT_Subset_r12          trpt_Subset_r12 optional
  };
  
  type record CommTx_ConfigInfo_Type {
    /* All fields are optional: present for first configuration omit means no change */
    CommResourcePoolList_Type  ResourcePoolList  optional,  /* Configure pool of resources for PSSCH and PSCCH */
    Mcs_Type                   Mcs_r12 optional,            /* Indicates the Modulation and Coding Scheme as defined in TS 36.212 [23, 14.2.1] */
    SL_SCH_Config_Type         SL_SCH_Config    optional,   /* SL-SCH/MAC configuration */
    STCH_ConfigList_Type       STCH_ConfigList  optional,   /* RLC PDCP */
    SSUE_SciConfig_Type        SL_Sheduling optional,       /* SCI fixed grant used */
    PscchConfig_Type           PscchConfig optional,
    PsschConfig_Type           PsschConfig optional
  };
  
  type record SSUE_SciConfig_Type {
    ImcsValue_Type             Imcs,              /* Modulation and coding scheme - 5 bits as defined in section 14.1.1 of [3] */
    SL_SciConfig_Type          SL_Sheduling       /* SCI fixed grant used */
  }
  
  type record SL_SCH_Config_Type {
    /* All fields are mandatory for initial configuration; omit means "keep as it is" */
    MAC_SLDCH_TestModeConfig_Type Mac  optional,
    B24_Type                      SS_UE_Layer2Id   optional
  };
  
  type union MAC_SLDCH_TestModeConfig_Type {
    Null_Type                  None,     /* MAC operation in normal mode for SL-DCH */
    Null_Type                  Config
  };
  
  type record length(1..tsc_MaxNumberSL_STCH) of STCH_Config_Type STCH_ConfigList_Type;
  const integer tsc_MaxNumberSL_STCH := 1;  /* Acc to 36.508 cl 4.2.2.3.1.1 */
  
  type record STCH_Config_Type {
    /* All fields are mandatory for initial configuration; omit means "keep as it is" */
    SLRB_Identity_Type     SLRB optional,
    SLRB_Config_Type       SL_RBIDConfig optional
  };
  
  type union SLRB_Config_Type {
    SLRB_ConfigInfo_Type       AddOrReconfigure, /* Add/re-configure SideLink RB */
    Null_Type                  Release           /* Release SideLink RB */
  };
  
  type record SLRB_ConfigInfo_Type {
    /* All fields are mandatory for initial configuration; omit means "keep as it is" */
    STCH_RLC_Type              Rlc     optional,
    SL_PDCP_Configuration_Type Pdcp    optional
  };
  
  type record STCH_RLC_Type {
    /* One RLC entity per Tx peer UE
       Uni-directional UM RLC
       sn-fieldLength = 5
       LCG = 3
       SN is set to the first received UMD PDU in Rx
       UM_Window size = 0
    */
    B24_Type                    Layer2Id  optional, /* Layer-2 Id of RLC entity: SourceLayer2-Id in Rx - DestinationLayer2Id in Tx */
    B5_Type                     LCID optional       /* Applied in Tx, in Rx SS shall use the value received in the first STCH data PDU */
  };
  
  type union SL_PDCP_Configuration_Type {
    Null_Type                   None,     /* no PDCP is configured; e.g. for SLRBs tested in MAC test cases */
    SL_PDCP_ConfigInfo_Type     Config
  };
  
  type record SL_PDCP_ConfigInfo_Type {
    PDCP_SLRB_Type              SlRb               optional, /* mandatory for initial configuration; omit means "keep as it is" */
    PDCP_TestModeConfig_Type    TestMode           optional  /* mandatory for initial configuration; omit means "keep as it is" */
  };
  
  type record PDCP_SLRB_Type  {
    /* PDCP SN length = 16b
       discard timer = undefined
       maxCID=15
    */
    SL_Rohc_Profiles_Type headerCompression optional /* mandatory for initial configuration; omit means "keep as it is" */
  };
  
  type  SL_PreconfigGeneral_r12.rohc_Profiles_r12 SL_Rohc_Profiles_Type;
  
  type record  PsbchConfig_Type {
    SideLinkChannelPower_Type       RelativeTxPower       optional /* Power for PSBCH relative to the SS-UE power level - value in dB */
  };
  
  type record  PsschConfig_Type {
    SideLinkChannelPower_Type       RelativeTxPower       optional /* Power for PSSCH relative to the SS-UE power level - value in dB */
  };
  
  type record  PscchConfig_Type {
    SideLinkChannelPower_Type       RelativeTxPower       optional /* Power for PSCCH relative to the SS-UE power level - value in dB */
  };
  
  } // End of Communication_Specific

  //----------------------------------------------------------------------------
  // Security types
  //----------------------------------------------------------------------------
  group SL_Security {
  
  type union SL_Security_Type {
    /* When not configured ciphering is considered as off  */
    SL_CipheringStartRestart_Type   StartRestart, /* Information to start/restart SL security protection in the PDCP SLRB */
    Null_Type                 Release             /* To release SL security protection in the PDCP */
  };
  
  type record SL_CipheringStartRestart_Type {
    /* Acc to 33.303
       NOTE:  Group Member Id (i.e. the Layer 2 source address of the sending UE) and Group Id (i.e. the destination Layer 2 identity of the group) are  configured in MAC and RLC (SL_SCH_Config_Type  and STCH_RLC_Type).
       Expiry time to be handled by TTCN
    */
    B24_Type     GroupId,       /* ProSe Layer-2 Group Id */
    B3_Type      AlgorithmInfo, /* Security algorithm acc to 33.303 Table E.5.2.2.7-1 */
    B256_Type    PGK,
    B64_Type     PGK_Id,        /* PGK identity */
    B16_Type     PTK_Id,        /* For SS_UE Tx - For PTK derivation at SS-UE side.
                                   The value may be different from UE PTK-Id */
    B16_Type     Counter        /* For SS_UE Tx - For PTK derivation */
  };
  
  } // End of SL_Security
  } // End of SL_SystemRequestAsp
  
  //****************************************************************************
  // Confirm
  //****************************************************************************
  group SL_SystemConfirmAsp {
  
  type record SL_CnfAspCommonPart_Type {
    SS_UE_Id_Type           SS_UE_Id,
    SL_RoutingInfo_Type     RoutingInfo,
    SL_TimingInfo_Type      TimingInfo,
    ConfirmationResult_Type Result
  };
  
  type union SL_SystemConfirm_Type {              /* Confirmations for SL system configuration;
                                                     in general to be sent after the configuration has been done
                                                  */
    Null_Type                   SS_UE_Config,     /* Confirmation for SS-UE configuration */
    Null_Type                   EnquireTiming,    /* Confirmation for EnquireTiming  */
    Null_Type                   L1MacIndCtrl,     /* Confirmation for L1MacIndCtrl  */
    Null_Type                   SL_Security       /* (no further parameters from SS) */
  };
  } // End of SL_SystemConfirmAsp
  
  //****************************************************************************
  // SL Indication
  //****************************************************************************
  group  SL_SystemIndicationAsp {
  
  type record SL_IndAspCommonPart_Type {
    SS_UE_Id_Type           SS_UE_Id,           /* Simulated SS-UE ID */
    SL_RoutingInfo_Type     RoutingInfo,        /* Channel ID or RB Id for STCH */
    SL_TimingInfo_Type      TimingInfo,         /* In coverage or Out of coverage timing */
    IndicationStatus_Type   Status
  };
  
  
  type union SL_SystemIndication_Type {
    charstring              Error,              /* Indicates an error situation in SS-UE;
                                                 * is not explicitly handled in TTCN but causes an INCONC due to default behaviour;
                                                 * an additional error code can be signalled in the common part of the ASP;
                                                 * SS shall raise an error in case of
                                                 * - Invalid TimingInfo
                                                 * - Data scheduled does not fit into an available resource in a resource pool
                                                 * (NOTE: additional cases may occur) */
    DiscSLSS_Ind_Type       DiscSync,           /* Indication for Discovery SLSS sent by UE
                                                 */
    CommSLSS_MIBSL_Ind_Type       CommSLSS_MIBSL      /* Indication for Communication SLSSID and MIB-SL sent by UE
                                                 */
  };
  
  type enumerated DiscSLSS_Ind_Type {
  /* When transmitted SLSS is associated with a PC5_DISCOVERY message according to RxDiscSyncRef.
     Start is triggered in SS_UE when
        - the first SLSS associated with a PC5_Discovery message is received by SS-UE since enabling of SLSS reporting
        - the previous SLSS indication was 'stop' or 'spurious' and an SLSS associated with a PC5_Discovery message is received by SS-UE

     Stop is triggered in SS_UE when
        - the first PC5_DISCOVERY message is received without SLSS
        - the previous SLSS indication was 'start' or 'spurious' and PC5_DISCOVERY message is received without SLSS

     Spurious is triggered in SS_UE when
        - the first SLSS is received without PC5_DISCOVERY message since enabling of SLSS reporting
        - the previous SLSS indication was 'start' or 'stop' and SLSS is received without associated PC5_DISCOVERY message
   */
    DiscSLSS_Start,
    DiscSLSS_Stop,
    DiscSLSS_Spurious
  };
  
  type union CommSLSS_MIBSL_Ind_Type {
  /* Start is triggered in SS_UE when
        - the first SLSS associated with a MIB-SL is received in the same subframe by SS-UE since enabling of reporting
        - the previous indication was 'stop' or 'spurious' and an SLSS associated with a MIB-SL is received by SS-UE

     Stop is triggered in SS_UE when
        - no SLSS/MIB-SL is transmitted in the period
        - the previous indication was 'start' or 'spurious' and no SLSS/MIB-SL is transmitted in the period

     Spurious is triggered in SS_UE when
        - the first SLSS is received without MIB-SL or MIB-SL is received without SLSS since enabling of SLSS reporting
        - the previous indication was 'start' or 'stop' and SLSS is received without MIB-SL or MIB-SL is received without SLSS
   */
    CommSyncParamInd_Type  CommSLSS_Start,
    Null_Type              CommSLSS_Stop,
    Null_Type              CommSLSS_Spurious
  };
  
  type record CommSyncParamInd_Type {
    SLSSID_r12                  SLSSID,
    MasterInformationBlock_SL   MIB_SL
  };
  
  } // End of SL_SystemIndicationAsp
  
  //****************************************************************************
  // SL System interface
  //****************************************************************************
  group SL_System_Interface {
  
  type record SL_SYSTEM_CTRL_REQ {
    /* ASP to request/control SideLink system configuration */
    SL_ReqAspCommonPart_Type       Common,  /* TimingInfo depends on respective primitive: */
    SL_SystemRequest_Type          Request  /* - SS_UE_Config
                                             *    TimingInfo: depends on the request
                                             * - EnquireTiming
                                             *    TimingInfo: 'now'
                                             * - L1MacIndCtrl
                                             *    TimingInfo: 'now'
                                             * - SL_Security
                                             *    RoutingInfo: SLRB
                                             *    TimingInfo: 'now'
                                             */
  };
  
  type record SL_SYSTEM_CTRL_CNF {
    /* ASP to confirm a SideLink system configuration request */
    SL_CnfAspCommonPart_Type       Common,  /* TimingInfo is ignored by TTCN (apart from EnquireTiming): SS may set TimingInfo to "None"
                                             * -EnquireTiming
                                             *    TimingInfo InCoverage: SFN/Subframe
                                             */
    SL_SystemConfirm_Type          Confirm
  };
  
  type record SL_SYSTEM_IND {
    /* ASP to report errors or system indications */
    SL_IndAspCommonPart_Type       Common,    /* The SS shall provide TimingInfo (SFN + subframe number) depending on the respective indication: */
    SL_SystemIndication_Type       Indication /* - Error
                                               *    TimingInfo: related to the error (if available)
                                               * - DiscSync
                                               *    TimingInfo: SFN/Subframe related to the SLSS
                                               * - CommSLSS
                                               *    TimingInfo: SFN/Subframe or DFN/DirectSubframe related to the SLSS
                                               * - CommMIB_SL
                                               *    TimingInfo: SFN/Subframe or DFN/DirectSubframe related to the MIB-SL
                                               */
  };
  
  type port SL_SYSTEM_PORT message {  /* SideLink UE PTC: Port for system configuration */
    out SL_SYSTEM_CTRL_REQ;
    in  SL_SYSTEM_CTRL_CNF;
  };
  
  type port SL_SYSIND_PORT message {  /* SideLink UE PTC: Port for system indications */
    in  SL_SYSTEM_IND;
  };
  
  } // End group SL_System_Interface
  
  } // End of SideLinkUE_Configuration
}

